NAG Library Routine Document

E04GYF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E04GYF is an easy-to-use quasi-Newton algorithm for finding an unconstrained minimum of a sum of squares of m nonlinear functions in n variables $(m \geq n)$. First derivatives are required.

It is intended for functions which are continuous and which have continuous first and second derivatives (although it will usually work even if the derivatives have occasional discontinuities).

2 Specification

```
SUBROUTINE E04GYF (M, N, LSFUN2, X, FSUMSQ, W, LW, IUSER, RUSER, IFAIL)

INTEGER             M, N, LW, IUSER(*), IFAIL
REAL (KIND=nag_wp) X(N), FSUMSQ, W(LW), RUSER(*)
EXTERNAL            LSFUN2
```

3 Description

E04GYF is similar to the subroutine LSFDQ2 in the NPL Algorithms Library. It is applicable to problems of the form

$$\text{Minimize } F(x) = \sum_{i=1}^{m} [f_i(x)]^2$$

where $x = (x_1, x_2, \ldots, x_n)^{\mathrm{T}}$ and $m \geq n$. (The functions $f_i(x)$ are often referred to as 'residuals'.) You must supply a subroutine to evaluate the residuals and their first derivatives at any point x.

Before attempting to minimize the sum of squares, the algorithm checks the subroutine for consistency. Then, from a starting point supplied by you, a sequence of points is generated which is intended to converge to a local minimum of the sum of squares. These points are generated using estimates of the curvature of $F(x)$.

4 References

Gill P E and Murray W (1978) Algorithms for the solution of the nonlinear least-squares problem *SIAM J. Numer. Anal.* **15** 977–992

5 Parameters

1: M – INTEGER *Input*

2: N – INTEGER *Input*

On entry: the number m of residuals, $f_i(x)$, and the number n of variables, x_j.

Constraint: $1 \leq \mathrm{N} \leq \mathrm{M}$.

3: LSFUN2 – SUBROUTINE, supplied by the user. *External Procedure*

You must supply this routine to calculate the vector of values $f_i(x)$ and the Jacobian matrix of first derivatives $\dfrac{\partial f_i}{\partial x_j}$ at any point x. It should be tested separately before being used in conjunction with E04GYF (see the E04 Chapter Introduction).

The specification of LSFUN2 is:

```
SUBROUTINE LSFUN2 (M, N, XC, FVEC, FJAC, LDFJAC, IUSER, RUSER)

INTEGER            M, N, LDFJAC, IUSER(*)
REAL (KIND=nag_wp) XC(N), FVEC(M), FJAC(LDFJAC,N), RUSER(*)
```

Important: the dimension declaration for FJAC must contain the variable LDFJAC, not an integer constant.

1: M – INTEGER *Input*

On entry: m, the numbers of residuals.

2: N – INTEGER *Input*

On entry: n, the numbers of variables.

3: XC(N) – REAL (KIND=nag_wp) array *Input*

On entry: the point x at which the values of the f_i and the $\dfrac{\partial f_i}{\partial x_j}$ are required.

4: FVEC(M) – REAL (KIND=nag_wp) array *Output*

On exit: FVEC(i) must contain the value of f_i at the point x, for $i = 1, 2, \ldots, m$.

5: FJAC(LDFJAC,N) – REAL (KIND=nag_wp) array *Output*

On exit: FJAC(i, j) must contain the value of $\dfrac{\partial f_i}{\partial x_j}$ at the point x, for $i = 1, 2, \ldots, m$ and $j = 1, 2, \ldots, n$.

6: LDFJAC – INTEGER *Input*

On entry: the first dimension of the array FJAC as declared in the (sub)program from which E04GYF is called.

7: IUSER($*$) – INTEGER array *User Workspace*

8: RUSER($*$) – REAL (KIND=nag_wp) array *User Workspace*

LSFUN2 is called with the parameters IUSER and RUSER as supplied to E04GYF. You are free to use the arrays IUSER and RUSER to supply information to LSFUN2 as an alternative to using COMMON global variables.

LSFUN2 must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)program from which E04GYF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

4: X(N) – REAL (KIND=nag_wp) array *Input/Output*

On entry: X(j) must be set to a guess at the jth component of the position of the minimum, for $j = 1, 2, \ldots, n$. The routine checks the first derivatives calculated by LSFUN2 at the starting point and so is more likely to detect an error in your routine if the initial X(j) are nonzero and mutually distinct.

On exit: the lowest point found during the calculations. Thus, if IFAIL $= 0$ on exit, X(j) is the jth component of the position of the minimum.

5: FSUMSQ – REAL (KIND=nag_wp) *Output*

On exit: the value of the sum of squares, $F(x)$, corresponding to the final point stored in X.

6: W(LW) – REAL (KIND=nag_wp) array *Workspace*
7: LW – INTEGER *Input*

On entry: the dimension of the array W as declared in the (sub)program from which E04GYF is called.

Constraints:

if $\mathrm{N} > 1$, $\mathrm{LW} \geq 8 \times \mathrm{N} + 2 \times \mathrm{N} \times \mathrm{N} + 2 \times \mathrm{M} \times \mathrm{N} + 3 \times \mathrm{M}$;
if $\mathrm{N} = 1$, $\mathrm{LW} \geq 11 + 5 \times \mathrm{M}$.

8: IUSER(∗) – INTEGER array *User Workspace*
9: RUSER(∗) – REAL (KIND=nag_wp) array *User Workspace*

IUSER and RUSER are not used by E04GYF, but are passed directly to LSFUN2 and may be used to pass information to this routine as an alternative to using COMMON global variables.

10: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if $\mathrm{IFAIL} \neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\mathrm{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\mathrm{IFAIL} = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: E04GYF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

$\mathrm{IFAIL} = 1$

On entry, $\mathrm{N} < 1$,
or $\mathrm{M} < \mathrm{N}$,
or $\mathrm{LW} < 8 \times \mathrm{N} + 2 \times \mathrm{N} \times \mathrm{N} + 2 \times \mathrm{M} \times \mathrm{N} + 3 \times \mathrm{M}$, when $\mathrm{N} > 1$,
or $\mathrm{LW} < 11 + 5 \times \mathrm{M}$, when $\mathrm{N} = 1$.

$\mathrm{IFAIL} = 2$

There have been $50 \times n$ calls of LSFUN2, yet the algorithm does not seem to have converged. This may be due to an awkward function or to a poor starting point, so it is worth restarting E04GYF from the final point held in X.

$\mathrm{IFAIL} = 3$

The final point does not satisfy the conditions for acceptance as a minimum, but no lower point could be found.

$\mathrm{IFAIL} = 4$

An auxiliary routine has been unable to complete a singular value decomposition in a reasonable number of sub-iterations.

IFAIL $= 5$
IFAIL $= 6$
IFAIL $= 7$
IFAIL $= 8$

There is some doubt about whether the point X found by E04GYF is a minimum of $F(x)$. The degree of confidence in the result decreases as IFAIL increases. Thus, when IFAIL $= 5$, it is probable that the final x gives a good estimate of the position of a minimum, but when IFAIL $= 8$ it is very unlikely that the routine has found a minimum.

IFAIL $= 9$

It is very likely that you have made an error in forming the derivatives $\dfrac{\partial f_i}{\partial x_j}$ in LSFUN2.

If you are not satisfied with the result (e.g., because IFAIL lies between 3 and 8), it is worth restarting the calculations from a different starting point (not the point at which the failure occurred) in order to avoid the region which caused the failure. Repeated failure may indicate some defect in the formulation of the problem.

7 Accuracy

If the problem is reasonably well scaled and a successful exit is made then, for a computer with a mantissa of t decimals, one would expect to get $t/2 - 1$ decimals accuracy in the components of x and between $t - 1$ (if $F(x)$ is of order 1 at the minimum) and $2t - 2$ (if $F(x)$ is close to zero at the minimum) decimals accuracy in $F(x)$.

8 Further Comments

The number of iterations required depends on the number of variables, the number of residuals and their behaviour, and the distance of the starting point from the solution. The number of multiplications performed per iteration of E04GYF varies, but for $m \gg n$ is approximately $n \times m^2 + O(n^3)$. In addition, each iteration makes at least one call of LSFUN2. So, unless the residuals and their derivatives can be evaluated very quickly, the run time will be dominated by the time spent in LSFUN2.

Ideally the problem should be scaled so that the minimum value of the sum of squares is in the range $(0, 1)$ and so that at points a unit distance away from the solution the sum of squares is approximately a unit value greater than at the minimum. It is unlikely that you will be able to follow these recommendations very closely, but it is worth trying (by guesswork), as sensible scaling will reduce the difficulty of the minimization problem, so that E04GYF will take less computer time.

When the sum of squares represents the goodness-of-fit of a nonlinear model to observed data, elements of the variance-covariance matrix of the estimated regression coefficients can be computed by a subsequent call to E04YCF, using information returned in segments of the workspace array W. See E04YCF for further details.

9 Example

This example finds the least squares estimates of x_1, x_2 and x_3 in the model

$$y = x_1 + \frac{t_1}{x_2 t_2 + x_3 t_3}$$

using the 15 sets of data given in the following table.

y	t_1	t_2	t_3
0.14	1.0	15.0	1.0
0.18	2.0	14.0	2.0
0.22	3.0	13.0	3.0
0.25	4.0	12.0	4.0
0.29	5.0	11.0	5.0
0.32	6.0	10.0	6.0
0.35	7.0	9.0	7.0
0.39	8.0	8.0	8.0
0.37	9.0	7.0	7.0
0.58	10.0	6.0	6.0
0.73	11.0	5.0	5.0
0.96	12.0	4.0	4.0
1.34	13.0	3.0	3.0
2.10	14.0	2.0	2.0
4.39	15.0	1.0	1.0

The program uses $(0.5, 1.0, 1.5)$ as the initial guess at the position of the minimum.

9.1 Program Text

```
!   E04GYF Example Program Text
!   Mark 24 Release. NAG Copyright 2012.
    Module e04gyfe_mod

!     E04GYF Example Program Module:
!            Parameters and User-defined Routines

!     .. Use Statements ..
      Use nag_library, Only: nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                  :: m = 15, n = 3, nin = 5,          &
                                             nout = 6, nt = 3
      Integer, Parameter                  :: lw = 8*n + 2*n*n + 2*m*n + 3*m
!     .. Local Arrays ..
      Real (Kind=nag_wp)                  :: t(m,nt), y(m)
    Contains
      Subroutine lsfun2(m,n,xc,fvec,fjac,ldfjac,iuser,ruser)
!       Routine to evaluate the residuals and their 1st derivatives.

!       .. Scalar Arguments ..
        Integer, Intent (In)                :: ldfjac, m, n
!       .. Array Arguments ..
        Real (Kind=nag_wp), Intent (Inout)  :: fjac(ldfjac,n), ruser(*)
        Real (Kind=nag_wp), Intent (Out)    :: fvec(m)
        Real (Kind=nag_wp), Intent (In)     :: xc(n)
        Integer, Intent (Inout)             :: iuser(*)
!       .. Local Scalars ..
        Real (Kind=nag_wp)                  :: denom, dummy
        Integer                             :: i
!       .. Executable Statements ..
        Do i = 1, m
          denom = xc(2)*t(i,2) + xc(3)*t(i,3)
          fvec(i) = xc(1) + t(i,1)/denom - y(i)
          fjac(i,1) = 1.0_nag_wp
          dummy = -1.0_nag_wp/(denom*denom)
          fjac(i,2) = t(i,1)*t(i,2)*dummy
          fjac(i,3) = t(i,1)*t(i,3)*dummy
        End Do

        Return
```

```
      End Subroutine lsfun2
    End Module e04gyfe_mod
    Program e04gyfe

!     E04GYF Example Main Program

!     .. Use Statements ..
      Use nag_library, Only: e04gyf, nag_wp
      Use e04gyfe_mod, Only: lsfun2, lw, m, n, nin, nout, nt, t, y
!     .. Implicit None Statement ..
      Implicit None
!     .. Local Scalars ..
      Real (Kind=nag_wp)                  :: fsumsq
      Integer                             :: i, ifail
!     .. Local Arrays ..
      Real (Kind=nag_wp)                  :: ruser(1), w(lw), x(n)
      Integer                             :: iuser(1)
!     .. Executable Statements ..
      Write (nout,*) 'E04GYF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

!     Observations of TJ (J = 1, 2, ..., nt) are held in T(I, J)
!     (I = 1, 2, ..., m)

      Do i = 1, m
        Read (nin,*) y(i), t(i,1:nt)
      End Do

      x(1:nt) = (/0.5_nag_wp,1.0_nag_wp,1.5_nag_wp/)

      ifail = -1
      Call e04gyf(m,n,lsfun2,x,fsumsq,w,lw,iuser,ruser,ifail)

      Select Case (ifail)
      Case (0,2:8,10:)
        Write (nout,*)
        Write (nout,99999) 'On exit, the sum of squares is', fsumsq
        Write (nout,99999) 'at the point', x(1:n)
      End Select

99999 Format (1X,A,3F12.4)
    End Program e04gyfe
```

9.2 Program Data

```
E04GYF Example Program Data
 0.14  1.0 15.0  1.0
 0.18  2.0 14.0  2.0
 0.22  3.0 13.0  3.0
 0.25  4.0 12.0  4.0
 0.29  5.0 11.0  5.0
 0.32  6.0 10.0  6.0
 0.35  7.0  9.0  7.0
 0.39  8.0  8.0  8.0
 0.37  9.0  7.0  7.0
 0.58 10.0  6.0  6.0
 0.73 11.0  5.0  5.0
 0.96 12.0  4.0  4.0
 1.34 13.0  3.0  3.0
 2.10 14.0  2.0  2.0
 4.39 15.0  1.0  1.0
```

9.3 Program Results

```
 E04GYF Example Program Results

 On exit, the sum of squares is      0.0082
 at the point      0.0824      1.1330      2.3437
```